


Simple matrix operations with SAS/IML

The goal of this exercise is simply to show you how to use SAS/IML as a matrix desk calculator. There is a collection of SAS/IML examples on the Psych 6140 web site, at <http://www.psych.yorku.ca/lab/psy6140/ex/iml.htm>. This tutorial is based on [imlmat4.sas](http://www.psych.yorku.ca/lab/psy6140/ex/imlmat4.sas). You can open this in the browser and copy/paste lines from any of these into SAS if you prefer. There is a SAS/IML Reference card at <http://www.psych.yorku.ca/lab/sas/SASIMLReferenceCard.pdf>.

1. Start SAS, then SAS/IML. Type and submit (F3 or ):

```
ods listing;
proc iml;
    reset print log fuzz fw=4;
```

The `reset` statement sets some convenient options to (a) `print` results automatically, (b) send printed output to the log window; (c) `fuzz` tiny numbers to zero; (d) use a field width =4 for printing matrices and vectors.

2. Define some matrices:

```
A = { 1 2, 3 4};
B = {1 1 1, 2 2 2};
C = {5 5, 6 6, 7 7};
```

3. Try some of the following expressions; Submit each set of lines and observe the result:

```
Result = A * B;
print (A * B);
Result = A * t(B);

BB = t(B) * B;
BB = B * t(B);

print C (t(C));
print C[rowname={R1 R2 R3} colname={C1 C2}];
```

4. A few more things to try, using subscripts and subscript operators:

```
x = { 1 2 4, 8 9 15,
      15 25 30, 12 16 9};

submat = x[1:3,];
submat = x[,1:3];

means = x[:,];
sums = x[+,];
sum = sum(x);
```

What do you think `s = x[,+];` would give? Try it. What about `s = x[+,+];`? Try it.

5. A couple of other functions: `t()`, `det()`, `inv()`

```
xt = t(x);
xpx = xt * x;
det = det(xpx);
inv = inv(xpx);
xx = xpx * inv;
```

6. Finally, there is a matrix library containing some handy additional functions:

SAS-IML

```
%include iml(matlib);  
d = dev(x);  
c = cov(x);  
r = corr(x);
```

When you want to end your use of SAS/IML, type:

```
quit;
```

7. Access the SAS/IML Help: Type `help iml` in the command box (upper left corner). The Language Reference section contains all details on SAS/IML operators, functions, and statements. There is also a handy [SAS IML Reference Card](#).
8. You might also like to try out some other examples from the course web page, e.g., [imlmat1.sas](#).